﻿/*
 * Copyright 2006-2007 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package asphalt2.util{
	import flash.utils.setTimeout;
	
	/**
	 * TaskChainで実行するタスク
	 * @author c9katayama
	 */
	internal class Task {
		
		internal var taskChain:TaskChain;
		//実行するFunction
		internal var executeFunction:Function;
		//次のFunctionを呼び出すインターバル
		internal var nextInterval:uint;
		//次に実行するTask
		internal var nextExecuteTask:Task;
		//executeFunctionが非同期かどうか
		internal var async:Boolean;
		
		internal var play:Boolean = true;
		
		public function execute():void{
			if(play){
				try{
					if(async){
						executeFunction(executeNext);
					}else{
						executeFunction();
						executeNext();
					}
				}catch(e:Error){
					taskChain.endTaskChain();
					throw e;
				}
			}
		}
		internal function executeNext():void{
			taskChain.endTask();
			if(nextExecuteTask != null){
				setTimeout(nextExecuteTask.execute,nextInterval);
			}
		}
	}
}